
伊藤 要二
アストラゼネカ株式会社

臨床統計･DM部 臨床統計グループ

その５．Model-based解析結果の要約
（MIXEDプロシジャを例として）

ODS機能による
MIXEDプロシジャ等の出力の改善

• 必要な情報だけを、見やすい形にレイア
ウトを加工して提示

• 検定結果をグラフ、集計表などとリンクさ
せる

• 複数モデルの情報を結合し、相互比較を
可能にする

MIXEDプロシジャなどはあらゆる解析に対応
することを目的としているため、冗長で見づ
らい場合が多い。

MIXEDプロシジャ等の問題点

ODS機能

必要な情報だけを見やすい形にレイアウトを
加工して提示する

例題
反復測定データの解析
反応変数: 収縮期血圧（SBP)
要因: 薬剤群、時点、薬剤群×時点、

薬剤群内被験者（変量効果）

SASプログラム
proc mixed data=sbp covtest;
  class group sub time;
  model sbp = group time group*time/
      ddfm=satterth;
  random
intercept/subject=sub(group);
run;

ODS機能によるHTMLへの出力

ods html body="c:¥Kproject¥model1.htm";
proc mixed data=sbp covtest;
  class group sub time;
  model sbp = group time group*time/
      ddfm=satterth;
  random intercept/subject=sub(group);
run;
ods html close;

出力オプションなしでの出力（１）

ModelInfo

ClassLevels

出力オプションなしでの出力（２）

Dimensions

IterHistory

ConvergenceStatus

出力オプションなしでの出力（３）

CovParms

FitStatistics

Tests3

NOITPRINT, NOCLPRINTオプション

ods html body="c:¥Kproject¥model1.htm";
proc mixed data=sbp covtest
      noitprint noclprint;
  class group sub time;
  model sbp = group time group*time/
      ddfm=satterth;
  random intercept/subject=sub(group);
run;
ods html close;

NOITPRINT, NOCLPRINTのみを指定し
た場合の最初の２つのテーブル

ModelInfo

Dimensions

MIXEDプロシジャから出力されるテーブルの例

○InfoCrit
○AsyCorr
○AsyCov

○Tests3
 ○FitStatistics
 ○CovParms

 (NOITPRINT)  ConvergenceStatus
○  IterHistory

○
(NOCLPRINT)

  ClassLevels

(NOINFO)  Dimensions
○  ModelInfo

optionにより削
除可(OPTION)

optionによ
り出力可 

常に出
力

テーブル名

ODS EXCLUDEによるテーブルの削除

ods html body='c:¥kproject¥excl.htm';

ods exclude dimensions;

proc mixed data=sbp
 ・
 ・
 ・
run;

ods output close;

出力すべきテーブルを指定

ods output ModelInfo     = modelinfo1
           Tests3        = tests31
           FitStatistics = fitstatistics1
           CovParms      = covparms1;
proc mixed data=sbp
 ・
 ・
 ・
run;
ods output close;

ModelInfoテーブルをHTMLに出力

ods html body="c:¥Kproject¥model1.htm";
data _null_;
  set modelinfo1;
  file print ods=(template="ModelInfo");
  put _ods_;
run;
・・・
ods html close; 

ModelInfoテーブルの例

変数名：　VALUE変数名：　DESCR

選択

ModelInfoテーブルの必要な行のみを
HTMLに出力

ods html body="c:¥Kproject¥model1.htm";
data _null_;
  set modelinfo1;
  file print ods=(template="ModelInfo");
  if descr in("Covariance Structure"
              "Estimation Method"
              "Degrees of Freedom Method");
  put _ods_;
run;
・・・
ods html close; 

必要な情報のみを残したModelInfo

18

必要な情報のみを残したFitStatistics

ods html body="c:¥Kproject¥model1.htm";
・
・
・
data _null_;

   set fitstatistics1;

   file print
ods=(template="FitStatistics");

   if descr in("-2 Res Log Likelihood"

                 "AIC (smaller is better)");

   put _ods_;

run;
・
・
・
ods html close;

COVTESTオプションを指定した場合の
CovParmsテーブル

CovParms Subject Estimate StdErr ZValue ProbZ

問題あり

CovParmsテーブル用テンプレートの作成
（変数ZValueおよびProbZを削除）

proc template;
  define table CovParms;
    parent=Stat.Mixed.CovParms;
    column covparm subject estimate stderr;
    header h1;
    define h1;
      text "Covariance parameter estimates";
    end;
  end;
run;

テンプレートを用いて
CovParmsテーブルを加工

ods html body="c:¥Kproject¥model1.htm";
 ・ ・ ・
data _null_;
  set covparms1;
  file print ods=(template="CovParms");
  put _ods_;
run;
 ・ ・ ・
ods html close;

22

CovParmsテーブル
（Z統計量およびｐ値を削除）

Tests3テーブルの出力

ods html body="c:¥Kproject¥model1.htm";
 ・ ・ ・
data _null_;

  set tests31;

  file print ods=(template="Tests3");

  put _ods_;

run;
 ・ ・ ・
ods html close;

24

MIXEDプロシジャから必要な情報のみを出力

Tests3テーブル （Type 3 F test)

26

平均値の推移グラフ

検定結果、グラフ、集計表のリンク

検定

集計表

グラフ

リンクリンク

28

共分散構造モデルの選択

• 変量効果
• 指数型系列相関
• 指数型系列相関＋測定誤差
• 変量効果＋指数型系列相関
• 変量効果＋指数型系列相関＋測定誤差

変量効果

yijk = µ + gi + sij + tk + (g×t)ik +  εijk

Var(sij)  =  σｓ
2,  Cov(sij, sij′) = 0

Var(εijk) = σε
2 , Cov(εijk, εijk′) = 0

∆t0

σε
2

σs
2

変量効果

proc mixed data=sbp;
  class group sub time;
  model sbp = group time group*time/
    ddfm=satterth;
  random
intercept/subject=sub(group);
 or
  random sub(group);
run;

∆t0

σe
2

指数型系列相関

yijk = µ + gi + tk + (g×t)ik + eijk

Cov(eijk, eijk′) = σe
2

 exp[−Δt(k, k′) /ρ]

指数型系列相関

proc mixed data=sbp;

  class group sub time;

  model sbp = group time group*time/

    ddfm=satterth;

  repeated time/ type=sp(exp)(time)

     subject=sub(group);

run;

yijk = µ + gi + tk + (g×t)ik + eijk +  εijk

Cov(eijk, eijk′) = σe
2

 exp[−Δt(k, k′) /ρ]

Var(εijk) = σε
2 ,  Cov(εijk, εijk′) = 0

指数型系列相関＋測定誤差

0 ∆t

σe
2

σε
2

指数型系列相関＋測定誤差

proc mixed data=sbp;

  class group sub time;

  model sbp = group time group*time/

ddfm=satterth;

  repeated time/type=sp(exp)(time)

subject=sub(group) local;

run;

変量効果＋指数型系列相関

yijk = µ + gi + sij + tk + (g×t)ik + eijk

Var(sij)  =  σｓ
2,  Cov(sij, sij′) = 0

Cov(eijk, eijk′) = σe
2

 exp[−Δt(k, k′) /ρ]

0 ∆t

σe
2

σs
2

変量効果＋指数型系列相関

proc mixed data=sbp;
  class group sub time;
  model sbp = group time group*time/
     ddfm=satterth;

  random
intercept/subject=sub(group);
  repeated time/ type=sp(exp)(time)
     subject=sub(group);

run;

yijk = µ + gi + sij + tk + (g×t)ik + eijk +  εijk

Var(sij)  =  σｓ
2,  Cov(sij, sij′) = 0

Cov(eijk, eijk′) = σe
2

 exp[−Δt(k, k′) /ρ]

Var(εijk) = σε
2 ,  Cov(εijk, εijk′) = 0

変量効果＋指数型系列相関＋測定誤差

0 ∆t

σε
2

σe
2

σs
2

変量効果＋指数型系列相関＋測定誤差

proc mixed data=sbp;
  class group sub time;
  model sbp = group time group*time/
     ddfm=satterth;

  random
intercept/subject=sub(group);
  repeated time/ type=sp(exp)(time)
     subject=sub(group) local;
run;

ODS　OUTPUTODS　OUTPUT

モデル適合性統計量FitStatisticsの併合

PROC MIXED
モデル１

FitStatistics

… PROC MIXED
モデル5

FitStatistics

SASデータセット

PROC MIXED
モデル2

FitStatistics

PROC TABULATE

併合

HTML

ODS　OUTPUT

ODS　HTML

40

モデル適合性統計量の比較

共分散の時間間隔に対するプロット

42

共分散構造の検討

相
互
参
照

共分散プロット モデル適合性統計量

モデル１ モデル2 モデル5…

まとめ
ODS機能によるMIXEDプロシジャ等の

出力の改善

• 必要な情報だけを、見やすい形にレイア
ウトを加工して提示

• 検定結果をグラフ、集計表などとリンクさ
せる

• 複数モデルの情報を結合し、相互比較を
可能にする


